

CHAPTER 9

EXTRA RESOURCES

Additional Resources

1. “The 14 Keywords” (<http://tiny.cc/14keywords/>): Learn about the 14 keywords and two operators in Small Basic.
2. “The If-ElseIf Ladder” (<http://tiny.cc/otherwiseif/>): Dig deep with a tutorial!
3. “Small Basic Reference Documentation: Shapes Object” (<http://tiny.cc/shapesobject/>): See all the methods for the Shapes object.

Review Questions

1. When should you use the If/ElseIf ladder?
2. What kind of operators are And and Or, and what do they do?
3. True or False: The And operator has higher precedence than the Or operator.

4. True or False: Enclosing the logical expressions in parentheses can change the order of evaluation in an If statement.
5. What is wrong with this code?

```
If (x < 10) Then
    y = 10
Else
    y = 30
ElseIf (x < 20) Then
    y = 20
EndIf
```

6. What's the output of the following code when score is 60? Try entering other values between one and 99.

```
If (score < 50) Then
    TextWindow.WriteLine("Red")
ElseIf (score < 75) Then
    TextWindow.WriteLine("Yellow")
ElseIf (score < 100) Then
    TextWindow.WriteLine("Green")
EndIf
```

7. What's the output of the following code when age is each of these values: 15, 30, and 40?

```
If ((age <= 18) Or (age >= 35)) Then
    TextWindow.WriteLine("Eligible for discount")
Else
    TextWindow.WriteLine("Pay full price")
EndIf
```

8. Write an If statement that sets the variable result to 1 when the variable answer has the value "y" or the value "Y".
9. Write an If statement that checks whether score is between 90 and 100 (inclusive) and displays the message Invalid if it isn't.
10. The following statement should check whether score is outside the range 0 to 100. What's wrong with this statement? How would you fix it?

```
If ((score < 0) And (score > 100)) Then
```

11. The following statement should check whether score is in the range 0 to 100. What's wrong with this statement? How would you fix it?

```
If ((score >= 0) Or (score <= 100)) Then
```

Practice Exercises

1. Do the next two blocks of code do the same thing? Explain your answer.

```
If (x <= 10) Then
    y = 5
ElseIf (x <= 20) Then
    y = 20
EndIf
```

```
If (x <= 10) Then
    y = 5
ElseIf (x > 10 And x <= 20) Then
    y = 20
EndIf
```

2. If x is 1, what's the result of each expression (true or false)?
 - a. (x <= 0) And (x > 0)
 - b. (x > 0) Or (x < 0)
 - c. (x <> 0) Or (x = 0)
 - d. (x >= 0) And (x < 0)
3. Write an If statement that sets y to 10 if the value stored in x is between 0 and 100 or less than -20.
4. If x is 1, y is 2, and z is 3, what's the result of these expressions?
 - a. (x < y) And (y < z)
 - b. (x < y) Or (y < z)
 - c. (x <> y) Or (y = z)
 - d. (x + y = z)
5. Write an If statement that sets x to 10 if y is greater than 50 or z is greater than 100, but not both. (Hint: you'll need to use both the And and Or operators.)
6. Write your own version of the Guess My Number game. First, have Small Basic select a random number between 1 and 100, and then ask the player to guess it. Instead of displaying "too high" or "too low", display a number of stars that shows the user how far the guess is from the secret number. Use this list:
 - a. If difference <= 2, display one star
 - b. If 2 < difference <= 4, display two stars
 - c. If 4 < difference <= 8, display three stars
 - d. If 8 < difference <= 16, display four stars

e. If $16 < \text{difference} \leq 32$, display five stars

f. Otherwise, display six stars.


7. How does Small Basic evaluate the following statement? What happens if you remove the parentheses around the compound condition $Y > 6$ Or $X < 0$?

```
If (X < 5 And (Y > 6 Or X < 0)) Then
```

8. Write a program that asks the user to enter the x- and y-coordinates of a point in a Cartesian plane. Then have the program display a descriptive message of the point's location as shown in the following examples:

(0, 0): origin
(3, 0): on the x-axis
(0, -2): on the y-axis
(1, 1): in the first quadrant
(-1, -2): in the third quadrant

9. Write a program that plays Craps. In this game, the player places a bet (in dollars) and then rolls a pair of dice. If the total is 7 or 11, the player loses the bet. If the total is 2, 3, or 12, the player wins the bet. Otherwise, the total is called the player's *point*. The player continues to roll the dice until he rolls his point again (and wins) or rolls 7 or 11 (and loses). The following flowchart explains the game.



10. The following program is a math quiz. Run the program and explain how it works.
-


```
Again:
num1 = Math.GetRandomNumber(10)
num2 = Math.GetRandomNumber(10)
op = Math.GetRandomNumber(4)

If (op = 1) Then      ' Addition
    msg = num1 + " + " + num2
    ans = num1 + num2
ElseIf (op = 2) Then ' Subtraction
    msg = num1 + " - " + num2
    ans = num1 - num2
ElseIf (op = 3) Then ' Multiplication
    msg = num1 + " * " + num2
    ans = num1 * num2
Else                 ' Division
    msg = (num1 * num2) + " / " + num2
    ans = num1
EndIf

TextWindow.Write(msg + " = ? ")
input = TextWindow.ReadNumber()
If (input = ans) Then
    TextWindow.WriteLine("Correct. Great job!")
Else
    TextWindow.WriteLine("Wrong. The answer is " + ans)
EndIf

TextWindow.WriteLine("")
Goto Again
```

11. Write a program that lets the player pick a secret number for the computer to guess. After making a guess, the computer should ask the player to enter 1, 2, or 3 depending on whether its guess was correct, too high, or too low, respectively.
12. Write a program to play Rock, Paper, Scissors against the computer. Your player makes their choice by entering 1, 2, or 3, whereas the computer makes a random choice. The rules are paper beats rock, rock beats scissors, and scissors beats paper!
13. Write a gambling game that uses dice. The player bets on a number from 1 to 6 and sets his bet amount (in dollars). The dealer then rolls three dice and counts how many dice match the number the player placed his bet on. The house pays 1:1 on a single match, 2:1 on a double match, and 5:1 on a triple match. If none of the dice matched the player's number, the player loses his bet. The following flowchart explains the game.



14. After his safe landing on Xenon, Captain Jack leaves his battleship to explore the planet. When he moves 40 meters from his ship, he sees a huge monster, about 50 meters away from the other side of the ship. Both Captain Jack and the monster start running toward the ship, but the monster's a little faster (the Captain's stride is 2 meters, whereas the monster's stride is 3 meters). At each step, Captain Jack must decide whether to run toward the ship or shoot the monster! When he fires his gun, he might hit the monster in a noncritical spot, seriously wound the monster, or miss him. Each time the monster's hit, he loses some energy and becomes weaker. If the monster loses all his energy, he'll die and the Captain will be safe. But if the monster gets to the ship first, the Captain will be stranded on the alien planet! Complete the following program by replacing the comments marked with * with your own code. Then play the game and save Captain Jack!

```

TextWindow.Title = "Space Monster"
energy = 100 ' Monster's energy
distC = 40 ' Captain Jack's distance to ship
distM = 50 ' Monster's distance to ship

Again:
TextWindow.Write("You are " + distC + " meters, and the monster ")
TextWindow.WriteLine("is " + distM + " meters away from the ship.")
TextWindow.WriteLine("The monster's energy is " + energy + ".")
TextWindow.Write("Choice: Move to ship (1) or Shoot (2)? ")
action = TextWindow.ReadNumber()

If (action = 2) Then ' Captain Jack wants to shoot
    hit = Math.GetRandomNumber(3) ' Three possibilities
    If (hit = 1) Then ' Monster is hit but not critically
        ' *Damage level between 1 and 4. Monster moves 2 meters.
    ElseIf (hit = 2) Then ' Critical hit
        ' *Damage level between 5 and 8. Monster moves 1 meter.
    Else ' Missed monster
        ' *Monster moves 3 meters.
    EndIf
ElseIf (action = 1) Then ' Captain Jack wants to move to ship
    ' *Captain Jack moves 2 meters and monster moves 3 meters.
EndIf

If (distC <= 0) Then ' Captain Jack is safe
    ' *Captain Jack reached his ship, and he is safe!
ElseIf (distM <= 0) Then ' Monster wins
    ' *Monster reached the ship first. Captain Jack is stranded!
ElseIf (energy <= 0) Then
    ' *Monster lost his energy. He's dead!
Else
    TextWindow.WriteLine("")
    Goto Again
EndIf

```
